

# EE230: Course Project

## Colour Sensor using Phase Sensitive Detection

Mohil Patel, 160070002  
Neharika Jali, 160040101

May 29, 2022

### 1 Goal

The goal of the project is to detect the colour of an object by detecting the light emitted by LEDs (of three different colours) that reflects from it using a photodiode. The current produced by the photodiode is then processed to detect the relative amounts of different colours using phase sensitive detection.

### 2 Simulation Setup

The entire setup includes 3LEDs-blue, green and red, a photodiode (which responds to visible light), a transimpedance amplifier, an inverting amplifier, a PSD unit and a low pass filter. The voltage input to the LEDs is a square wave with different frequencies for different colours. The PSD Unit of the different colours are controlled by their respective frequencies.

The light from the LED falls on the reverse biased photodiode and generates current. During simulation we replaced this with a voltage source. Simultaneous light from 3 LEDs is modelled by using 3 voltage sources in series. This voltage is amplified by the inverting amplifier and then fed to the PSD unit.

The PSD unit for each colour detects signals of only that particular colour's LED. This specificity is achieved by using the same frequency signals for LED control and PSD switch control. In each of the phase sensitive detection units, capacitor controlled by control 1 gets charged when the LED is on, thus recording information about signal and noise. The capacitor connected to control ( not Control 1) gets charged when the LED is off, thus recording information about only the noise. The subtractor circuit which comes after this subtracts noise from (signal+noise) and gives the output.

The output of the PSD when passed through a low pass filter gives a DC value as the output. The DC values of each of the parts (red, blue, green) are collected and then processed to find the relative amounts of each colour in the given object.

We simulated the Trans-impedance amplifier, Inverting Amplifier and PSD Circuit with 3 different controls corresponding to red, green and blue light.




Figure 1: Transimpedance Amplifier circuit




Figure 2: Inverting Amplifier




Figure 3: Complete PSD Circuit

### 3 Simulation Results

#### 3.1 Trans-impedance Amplifier

Input current to trans-impedance Amplifier is  $30\mu\text{A}$  and the corresponding output is as shown in figure 5.




Figure 4: Input Current to Trans-impedance Amplifier




Figure 5: Output of Trans-impedance Amplifier

## 3.2 Inverting Amplifier

Input to inverting amplifier is 0.3V square wave pulse and output is clearly the inverted output with gain of 10.




Figure 6: Input and Output of Inverting Amplifier

## 3.3 PSD Circuit

In the PSD Circuit we have 3 voltage sources corresponding red, green and blue led signals which are square pulses shown equivalently as  $V_r$ ,  $V_g$  and  $V_b$ .




Figure 7: Output for  $V_r = 5V$ ,  $V_g = 0V$  and  $V_b = 0V$




Figure 8: Output for  $V_r = 3V$ ,  $V_g = 0V$  and  $V_b = 0V$




Figure 9: Output for  $V_r = 5V$ ,  $V_g = 5V$  and  $V_b = 0V$




Figure 10: Output for  $V_r = 5V$ ,  $V_g = 5V$  and  $V_b = 5V$




Figure 11: Output for  $V_r = 5V$ ,  $V_g = 3V$  and  $V_b = 1V$

## 4 Observations and Inferences

Observations will be sectioned in 3 different each showing observations of the 3 subsections of in the above section.

Firstly, Trans-impedance Amplifier. As it can be seen that the trans-impedance circuit clearly amplifies the input current signal of  $30\mu A$  to  $3V$ . As we know that the output of photo-diode is a current in range of micro-amperes while in ON state so we can say that the circuit is working satisfying and is producing output in readable range.

Now, inverting amplifier this amplifier is used specifically as to produce output in a rather more readable range of voltage as the output of trans-impedance amplifier will be milli-volts if current is  $3\mu A$ , so using a inverting amplifier we are producing output in volts range. This depends on IC used and light intensity and will be re-calibrated while making the circuit, if needed. the simulation result is pretty simple and can be seen in fig 6. Its an inverting amplifier with gain of 10.

Lastly and most important, we have PSD circuit. Now in these circuit we will be inputting the output of inverting amplifier, which as we know



is a square wave signal, so we can simulate the signal with voltage sources. Now as red, green and blue is having different frequencies so we will have square wave voltage sources with different frequencies. Also there will be 3 PSD circuits with different controls and they will differentiate between the 3 different signals. Now for inferences lets first compare fig 7 and 8. As we can clearly understand the output of the circuit if only single led is blinking will have only that PSD circuit working, giving some output and other two will have 0 output. Now also the output depends on the level of the voltage at inverting amplifier output so we can say that higher current will produce higher voltages and thus the output will be higher. Now lets see figure 9 and 10 in this two figures it can be seen that the output of PSD circuit corresponding to the colors, if they are producing same current, will be same. So if blue is not producing any current and red and green are producing currents if same intensity then the output will have red and green circuit's PSD as some voltage which is same voltage and the blue circuit will have zero output. Similar is for the next figure. Finally we will be seeing figure 11, this figure shows that the circuit will have different output voltage values if the input current produced by the lights are different so we will be able to get percentage red, green and blue as the output.

## 5 Appendix

### 5.1 Trans-impedance Amplifier Netlist

Trans-impedance amplifier

```
.include ua741.txt

R1 1 2 10k
R2 2 3 100k

xs 0 2 4 5 3 ua741

Vcc 4 0 15V
Vss 0 5 15V
i1 0 1 pulse(0 30u 0 5n 5n 0.5m 1m)
```

```
.control
tran 0.01m 10m
plot (V(1)-V(2))/10k
plot V(3)
.endc
.end
```

## 5.2 Inverting Amplifier Netlist

Inverting Amplifier

```
.include ua741.txt

R1 3 4 1k
R2 4 5 10k

xs 0 4 6 7 5 ua741

Vcc 6 0 15V
Vss 0 7 15V
Vin 3 0 pulse(0 0.3 0 5n 5n 0.5m 1m)

.control
tran 0.01m 10m
plot V(3) V(5)
.endc
.end
```

## 5.3 PSD Circuit Netlist [2]

PSD\_Circuit

```
.include ua741.txt
.include CD4066.txt

Vp 16 0 15V
Vn 0 17 15V
Vsmallp 18 0 5V
```

xs\_1a 15a 5 9a 18 0 CD4066  
xs\_2a 14a 5 8a 18 0 CD4066

C1a 9a 0 1u  
C2a 8a 0 1u  
R1a 9a 0 100k  
R2a 8a 0 100k

R3a 9a 10a 10k  
R4a 8a 11a 10k  
R5a 11a 0 10k  
R6a 10a 12a 10k

xa 11a 10a 16 17 12a ua741

R7a 12a r 100k  
C3a r 0 1u

xs\_1b 15b 5 9b 18 0 CD4066  
xs\_2b 14b 5 8b 18 0 CD4066

C1b 9b 0 1u  
C2b 8b 0 1u  
R1b 9b 0 100k  
R2b 8b 0 100k

R3b 9b 10b 10k  
R4b 8b 11b 10k  
R5b 11b 0 10k  
R6b 10b 12b 10k

xb 11b 10b 16 17 12b ua741

R7b 12b g 100k  
C3b g 0 1u

```
xs_1c 15c 5 9c 18 0 CD4066
xs_2c 14c 5 8c 18 0 CD4066
```

```
C1c 9c 0 1u
C2c 8c 0 1u
R1c 9c 0 100k
R2c 8c 0 100k
```

```
R3c 9c 10c 10k
R4c 8c 11c 10k
R5c 11c 0 10k
R6c 10c 12c 10k
```

```
xc 11c 10c 16 17 12c ua741
```

```
R7c 12c b 100k
C3c b 0 1u
```

```
Vc1a 14a 0 pulse(0 5V 0 5n 5n 0.5m 1m)
Vc2a 15a 0 pulse(0 5V 0.5m 5n 5n 0.5m 1m)
Vc1b 14b 0 pulse(0 5V 0 5n 5n 0.33m 0.66m)
Vc2b 15b 0 pulse(0 5V 0.33m 5n 5n 0.33m 0.66m)
Vc1c 14c 0 pulse(0 5V 0 5n 5n 0.25m 0.5m)
Vc2c 15c 0 pulse(0 5V 0.25m 5n 5n 0.25m 0.5m)
```

--Pulse voltage Vina, Vinb and Vinc can be changed to change input  
--voltage levels of red, green and blue respectively.

```
Vina 5a 0 pulse(0 5V 0 5n 5n 0.5m 1m)
Vinb 5b 5a pulse(0 5V 0 5n 5n 0.33m 0.66m)
Vinc 5 5b pulse(0 0V 0 5n 5n 0.25m 0.5m)
```

```
.control
tran 0.1m 300m
plot V(r) V(g) V(b)
.endc
.end
```

## 5.4 UA741 Model Netlist [3]

```
*-----  
*  
* To use a subcircuit, the name must begin with 'X'. For example:  
* X1 1 2 3 4 5 ua741  
*  
* connections:   non-inverting input  
*                | inverting input  
*                | | positive power supply  
*                | | | negative power supply  
*                | | | | output  
*                | | | | |  
.subckt ua741    1 2 3 4 5  
*  
  c1  11 12 8.661E-12  
  c2   6  7 30.00E-12  
  dc   5 53 dx  
  de  54  5 dx  
  dlp 90 91 dx  
  dln 92 90 dx  
  dp   4  3 dx  
  egnd 99  0 poly(2) (3,0) (4,0) 0 .5 .5  
  fb   7 99 poly(5) vb vc ve vlp vln 0 10.61E6 -10E6 10E6 10E6 -10E6  
  ga   6  0 11 12 188.5E-6  
  gcm  0  6 10 99 5.961E-9  
  iee 10  4 dc 15.16E-6  
  hlim 90  0 vlim 1K  
  q1  11  2 13 qx  
  q2  12  1 14 qx  
  r2   6  9 100.0E3  
  rc1  3 11 5.305E3  
  rc2  3 12 5.305E3  
  re1 13 10 1.836E3  
  re2 14 10 1.836E3  
  ree 10 99 13.19E6  
  ro1  8  5 50  
  ro2  7 99 100
```

```

rp    3  4 18.16E3
vb    9  0 dc 0
vc    3 53 dc 1
ve   54  4 dc 1
vlim  7  8 dc 0
vlp   91  0 dc 40
vln   0 92 dc 40
.model dx D(Is=800.0E-18 Rs=1)
.model qx NPN(Is=800.0E-18 Bf=93.75)
.ends

```

## 5.5 CD4066 Model [1]

```

* CD4066 Analog Switch
* SYM=CD4066
* Transistor models are from LTspice group member kcin_melnick
* See message number 16897, http://tech.groups.yahoo.com/group/LTspice/
* Analog Switch Control In Out Vdd Vss
.SUBCKT CD4066 2 11 4 10 7
X1 2 6 10 7 INVERT
X2 6 1 10 7 INVERT
M1 14 6 7 7 CD4007N
M7 11 6 14 10 CD4007P
M3 11 1 14 14 CD4007N
M4 11 1 4 14 CD4007N
M8 11 6 4 10 CD4007P
.SUBCKT INVERT 1 2 3 4
* Inverter In Out Vcc Vss
M1 2 1 3 3 CD4007P
M2 2 1 4 4 CD4007N
.ENDS
.MODEL CD4007N NMOS (
+ LEVEL=1 VTO=1.44 KP=320u L=10u W=30u GAMMA=0 PHI=.6 LAMBDA=10m
+ RD=23.2 RS=90.1 IS=16.64p CBD=2.0p CBS=2.0p CGS0=0.1p CGD0=0.1p
+ PB=.8 TOX=1200n)

.MODEL CD4007P PMOS (

```

```
+ LEVEL=1 VTO=-1.2 KP=110u L=10U W=60U GAMMA=0 PHI=.6 LAMBDA=40m
+ RD=21.2 RS=62.2 IS=16.64P CBD=4.0P CBS=4.0P CGSO=0.2P CGDO=0.2P
+ PB=.8 TOX=1200N)
.ENDS
```

## References

- [1] <http://www.embeddedcomponents.com/blogs/2011/12/cd4066-ltspice-simulation/>
- [2] How Dark Is Dark, Abhishek Ashtekar, WEL, Department of Electrical Engineering, IIT Bombay
- [3] EE236, WEL, Department of Electrical Engineering, IIT Bombay